Phosphorous (P) is an essential nutrient for life. It is found in the RNA and DNA of all organisms, as well as in the adenosine triphosphate molecule (ATP) that allows cells to carry out metabolic processes. In vertebrates, P is an important component of bones and teeth. P is found in every part of the Earth system, but in vastly different amounts. Most P on Earth is found in the +5 valence state (able to accept 5 electrons) as \( \text{PO}_4^{3-} \), commonly known as phosphate. Phosphate is found dissolved in water, found as the mineral apatite (nearly 95% of P in minerals is found in apatite) in soils, rocks, teeth, and bones, and found as mineral dust in the atmosphere.

P is unusual among many of the elements necessary for life in that it does not have a naturally occurring gaseous form of any significance. For this reason, the cycling of P throughout the Earth system is dependent on slow-acting geologic processes such as weathering of rocks, transport of sediments in streams, decay of terrestrial plants, and upwelling of ocean water. Because these processes occur slowly, P is considered a limiting nutrient. To understand what this means, consider the following: Organisms have a tendency toward exponential growth given infinite resources. However, since P is only made available to organisms a little at a time, it acts as a brake on this exponential growth. You might be interested to know that the tremendous increase in the human population that has occurred in the last century is a by-product primarily of the use of phosphate fertilizers on crops. More phosphate for crops means bigger harvests, and bigger harvests allow the human population to grow. The fertilizer phosphate has been mined from natural accumulations of phosphorous rich sediments deposited in the oceans and on land (bat guano and bird guano are the terrestrial sources). At present rates of consumption, we have enough minable phosphate to last us about another 400 years. What do you think might happen to the future population of the Earth when all of the minable phosphate has been exhausted?

Today we will create a STELLA model of the global phosphorous cycle and use it to test a number of scenarios. As we use the model, consider how you might be able to extend it to examine the influence of the P cycle on various ecologic systems. I want you to answer all of the questions below. Open up Microsoft Word at the same time that you have STELLA running. You can write the answers to your questions as you go along. You can also paste directly into your Word document any graphs you would like to use to answer your questions.
Readings


![Stella model diagram](image-url)

The above figure is from Chameides and Perdue, 1997, and is the basis for the P cycle modeling exercises. It represents a Phosphorus cycle in steady state. The STELLA model
created to replicate this cycle looks nearly identical, but contains additional arrows that link the contents of each reservoir to the outflows so that non-steady state behavior can be explored (see STELLA model graphic below). The use of the pink linking arrows is explained below.

**Exercises**
1) Using the Chameides and Perdue (1997) figure 5.5 on pg. 103, create a STELLA model of the global phosphorus cycle. For the moment, ignore the minable P part of the sediments reservoir.

**The Biogeochemical Phosphorus Cycle**

![Biogeochemical Phosphorus Cycle Diagram](image)

**STELLA code - provided in the event that you are using an older version of STELLA than that we're using or if you have problems downloading and opening the model**

```stella
deep_ocean(t) = deep_ocean(t - dt) + (sinking_and_decay + mixing - sedimentation - upwelling) * dt
INIT deep_ocean = 1.e5
INFLOWS:
sinking_and_decay = ocean_biota*(40./44.)
mixing = surface_ocean*(18./2800.)
```
OUTFLOWS:
sedimentation = deep_ocean*(2./1.e5)
upwelling = deep_ocean*(56./1.e5)

land_biot(a(t) = land_biot(a(t - dt) + (terrestrial_production - decay) * dt
INIT land_biot(a = 2600
INFLOWS:
terrestrial_production = soils*(310./2.e5)
OUTFLOWS:
decay = land_biot(a*(310./2600.)

ocean_biot(a(t) = ocean_biot(a(t - dt) + (oceanic_production - sinking_and_decay - ocean_decay) * dt
INIT ocean_biot(a = 44
INFLOWS:
oceanic_production = surface_ocean*(980./2800.)
OUTFLOWS:
sinking_and_decay = ocean_biot(a*(40./44.)
ocean_decay = ocean_biot(a*(940./44.)

sediments(t) = sediments(t - dt) + (weathering_insoluble + sedimentation - uplift) * dt
INIT sediments = 2.e9
INFLOWS:
weathering_insoluble = soils*(18./2.e5)
sedimentation = deep_ocean*(2./1.e5)
OUTFLOWS:
uplift = sediments*(20./2.e9)

soils(t) = soils(t - dt) + (decay + uplift - weathering_insoluble - weathering_soluble -
terrestrial_production) * dt
INIT soils = 2.e5
INFLOWS:
decay = land_biot(a*(310./2600.)
uplift = sediments*(20./2.e9)
OUTFLOWS:
weathering_insoluble = soils*(18./2.e5)
weathering_soluble = soils*(2./2.e5)
terrestrial_production = soils*(310./2.e5)

surface_ocean(t) = surface_ocean(t - dt) + (upwelling + weathering_soluble + ocean_decay - mixing -
oceanic_production) * dt
INIT surface_ocean = 2800
INFLOWS:
upwelling = deep_ocean*(56./1.e5)
weathering_soluble = soils*(2./2.e5)
ocean_decay = ocean_biot(a*(940./44.)
OUTFLOWS:
mixing = surface_ocean*(18./2800.)
oceanic_production = surface_ocean*(980./2800.)

As you set up your model think about the fluxes between the reservoirs. Is this a
closed system or open system model? How should the fluxes be specified?
This is a closed system model, in which the fluxes from one reservoir to the next depend on the quantity of phosphorus contained in the reservoirs from which the fluxes emanate. Chameides and Perdue cast this relationship mathematically as Flux = transfer coefficient * Reservoir contents, where the transfer coefficient is equivalent to the flux value divided by the reservoir contents when the entire system is at steady state. For example, in using Chameides and Perdue's Fig. 5.5 above, the transfer coefficient associated with the terrestrial production flux would measure 310 Tg/year divided by 200,000 Tg.

In STELLA, the pink linking arrow is used to show the dependence of the outflow from each reservoir on the amount of material contained within the reservoir. Once the pink linking arrow is applied, the reservoir becomes a selectable quantity in the flux specification dialog box.

Fluxes should be specified as the reservoir value multiplied by the transfer coefficient, which yields the original steady state flux value when the system is at steady state. For example, terrestrial production = soils * (310./2.e5), where 310 is the steady state terrestrial production flux and 2.e5 is the steady state value of P content in the soils reservoir, yields a value of 310 for terrestrial production when the soils reservoir contains its steady state value of 2.e5. The reservoir values will be modified in future experiments, interrupting the steady state.

2) Once your model is created, run it and keep track of the sizes of all of the reservoirs over time. Use a simulation period of 10 years and a timestep of 0.25. Describe what you see.

- Range = 0-10 years, Timestep (DT) = 1/4 (0.25) years
Running the model with the above parameters results in the following graphical behavior of the reservoirs:
The model is supposed to show steady state behavior. Does it? Explain how you know.

- None of the reservoirs appear to be in a steady state, as they are exhibiting change in quantity over time. The reservoir values should plot as horizontal lines, revealing no change over time.

If it doesn't, try to explain why it's not in steady state and modify your model until you get it to reach steady state.

The behavior shown in the graph above is a result of the value of the time step (DT). The model shows steady state behavior with a time step of 1/32 year or lower (see graph below). The reason the model doesn't come to steady state with a longer time step is related to the size of the Ocean Biota reservoir compared to the decay flux. The flow out of the Ocean Biota reservoir via decay is 940 Tg/yr but the value in Ocean Biota is only 44 Tg. With too long a time step, the model tries to remove more material in each iteration than is present in the reservoir, resulting in pathological behavior. When the time step length is reduced, less material is removed than is in the reservoir and the model can reach steady state. This problem illustrates the 1/2 DT rule, which states that the model time step should be halved until such time as the results between successive model runs are identical. Once identical results have been achieved, one can select the longer of the 2 time steps to proceed with experiments.

The help documentation that comes with the STELLA software gives an excellent introduction to time steps and how to choose their length.

-Range = 0-10 years, DT = 1/32 year
3) Now we'll explore an anthropogenic change to the system. Let's see what happens if we take all of the minable phosphorous in the world and apply it as fertilizer to the land surface.

a) Which reservoir do we change, and by how much?

- This experiment involves changing the Soils reservoir by adding the 10,000 Tg of minable phosphorus to it shown in Fig. 5.5 of Chameides and Perdue. This results in a new quantity of 210,000 Tg for Soils.

b) Run the model again and both **describe and explain** what you see. You may want to use a longer simulation time (try 100 years).

- Range = 0-100 years, DT = 1/32 year

- The system does not reach steady after running for 100 years. The graph reveals the gradual depletion of the Soils reservoir from its new, higher value of 210,000 Tg and the steady increase in the Deep Ocean, Sediments and Ocean Biota Reservoirs produced as the increased P is passed from the Soils to the other reservoirs. The Land Biota reservoir increases drastically in response to the higher value of P in the Soil. However, the Land Biota curve flattens out and slowly begins to decline after about 25 years, and if we were able to run the model for a longer time period, we would find that the entire system would adjust itself to a new steady state with somewhat higher values of P in each reservoir than in the previous experiment. Unfortunately, STELLA is limited in the total number of iterations it is able to carry out, and even if the model is run for 1000 years, the new steady state is not reached. This is not a problem with the Fortran model, which can be run until steady state is achieved if so desired.
4) In the 1950s to 1970s people began to realize that over-application of fertilizer and use of phosphate-containing detergents was leading to serious declines in the health of aquatic ecosystems. The influx of P caused algal blooms in lakes and along coasts. As the algae died, they fell through the water column and decayed, a process that consumed dissolved oxygen in the water as carbon compounds were oxidized to CO$_2$. The loss of dissolved oxygen prompted fish kills and the death of other aquatic animals that use gills to breathe.

Sketch out a STELLA model that you could add to your existing model that would simulate this process (called eutrophication). I don't expect you to come up with numbers. Just show me the fluxes and reservoirs you would have to consider to model this problem. One of you may want to use this as your independent research project later in the semester!

**Eutrophication Model Example**
- The most common problem students have with creating a simple sketch model such as this is that they have a tendency to try to link directly reservoirs containing different quantities. For example, linking the algal blooms reservoir directly to the fish reservoir. While STELLA allows this and it's possible to include unit conversions that would allow such a model to be valid, I try to discourage this in order to get students to consider carefully all of the different processes in the modeling problem and to get them to think about the units they're using. For example, algae don't directly lead to fish death. Instead, decay of algae leads to a decline in dissolved oxygen in the water, and this decline leads to fish death. Linking algae to fish thus requires an intermediary step in which oxygen content of water is modeled.

5) Set the reservoirs back to their initial values in problem 1. We will now explore a possible impact of a climatic change on the P cycle. Today, ocean circulation is considered akin to a huge conveyor belt. Warm water near the equator in the Atlantic flows northward toward Scandinavia. As it flows, evaporation causes it to become more saline, and heat exchange with the atmosphere causes it to cool off. Both of these processes cause the water to increase in density, and once it reaches the North Atlantic, the water sinking, flowing back across the bottom of the ocean as North Atlantic Deep Water. During the Last Glacial Maximum, about 20,000 years ago, this thermohaline circulation, as it is called, is thought to have either stopped or slowed down significantly. The probable cause: melting of ice from the large continental ice sheets around the North Atlantic decreased the salinity of the water, making it less dense and causing the sinking rate to decline or go to zero. Since water no longer was sinking in the N. Atlantic, the entire conveyor belt circulation shut down, depriving northern Europe of an important source of heat (see Bradley, R.S., 1999, Paleoclimatology: Reconstructing Climates of the Quaternary, 2nd edition, New York, Academic Press, p. 260-275 for a nice summary of this problem). This hypothesis has not been fully proven yet, but ocean circulation models suggest that this mechanism could have worked.

Let's see what kind of impact we might expect on the P cycle, were the ocean to no longer circulate.

a) What do we need to change about the model?

- This change involves removing the mixing and upwelling flows between the Surface Ocean and Deep Ocean reservoirs. This can be accomplished by simply multiplying the value of these flows by zero. In this way, the model structure remains intact.

b) Once you've made your change, run the model for a 1000-year period. What happens to the different reservoirs over time? Describe and explain what you see.
- Range = 1-1000 years, DT = 1/32 year

- Removing the flows of mixing and upwelling results in the graphs of each reservoir changing as the model adjusts itself toward a new steady state. The ocean biota and surface ocean reservoirs show behavior similar to exponential decay, declining quickly at first, and leveling out eventually at new levels reflecting a balance between inflow of P via soluble weathering and outflow of P sinking and decay of ocean biota.

Since upwelling no longer removes P from the deep ocean reservoir, the values in it, the sediments, soils and land biota all increase as the extra P propagates
through the system. The land biota and soils reservoirs exhibit patterns closely resembling exponential growth, but a glance at the y-axis scale reveals that they are in fact changing very little, which is to be expected since the P coming to them has to pass through the enormous Sediments reservoir. Because this reservoir is so large, it is not easily perturbed and acts as a buffer for changes in other parts of the system.

c) Which reservoirs adjust most dramatically and why? To answer this, pay attention to the scale of each variable on your graph.

- The Surface Ocean and Deep Ocean both exhibit the most dramatic changes. This makes sense, as the major inflows and outflows to and from these reservoirs have been cut off. Ocean Biota also drops significantly (from 50 to around 5), because the major flow into Ocean Biota is from Surface Ocean, which has changed fairly dramatically.

d) Which reservoirs adjust most gradually and why?

- The Sediments reservoir adjusts most gradually to the perturbation. This is because the influx into Sediments from Deep Ocean is very small compared to the size of the reservoir. The Soil and Land Biota reservoirs also adjust slowly. This is because in order to be affected by a change in Surface or Deep Ocean, the phosphorus has to first pass through Sediments, which adjusts very gradually.

e) Why does the deep ocean reservoir first increase in size and then very gradually begin to decrease? Would you have predicted this behavior?

- The initial large increase in quantity of the Deep Ocean occurs because the major outflow from the reservoir (upwelling) has been cut off, allowing P coming in from Ocean Biota to build up fairly quickly. Eventually however, the inflow from Ocean Biota drops off as a result of the decrease in Surface Ocean P, and the Deep Ocean begins to lose more from sedimentation than it gains from sinking and decay.

6) Next we'll study what would happen if the Earth underwent an increase in plate tectonic rates, such as is thought to have occurred during the late Cretaceous. Set your model back to the way it was in problem 1.

a) Suppose plate tectonic rates increased so that continents collided at faster rates, resulting in an increase in the uplift transfer coefficient of 25%. What do you predict would happen to the amount of P in all of the reservoirs?
- The correct prediction is that the quantity of Phosphorous in each reservoir will increase except for the Sediments reservoir, which should decline.

b) Change your model and run it for 1000 years. Were you correct in your predictions? Why or why not?

- Range = 1-1000 years, DT = 1/32 year

- The Sediments reservoir decreases because sediment is being removed at a faster rate than the previous steady state rate. The phosphorous removed from
The sediment is then distributed throughout the other reservoirs, causing them to increase.

c) Keep track of the uplift and insoluble weathering fluxes as a function of time. **Describe and explain** their behavior.

- Range = 1-1000 years, $DT = 1/32$ year

- Over time, the uplift flow is decreasing because, although the rate of removal has increased, the volume of $P$ in the sediments reservoir is decreasing. The insoluble weathering flow increases because of an increase in $P$ in the soils reservoir.

7) Another by-product of glaciation is that the rate of insoluble weathering probably went up. Glaciers produce vast quantities of finely ground up rock flour that is transported by streams to the oceans. Set your model back to the way it was in problem 1. Now, modify it to exhibit a 25% increase in the transfer coefficient for insoluble weathering.

**Describe and explain** what you see as you run the model.
- Range = 1-1000 years, DT = 1/32 year

- Range = 1-1000 years, DT = 1/32 year

- After increasing the insoluble weathering coefficient by 25%, all of the Phosphorous levels of the reservoirs decrease over time except for the Sediments reservoir, which increases linearly. This behavior is a result of the depletion of the Soil reservoir by the increased outflow. This increased depletion causes all of the other reservoirs dependent on it to be depleted as well.

8) Land plants first arose during the Silurian (439-409 Ma). Let's do an experiment in which we look at the phosphorous cycle in the absence of land plants. Set your model back to the steady state scenario in problem 1.
a) What do we have to change about the model to simulate no land plants?

- This change involves removing the effect of the Land Biota reservoir and its flows. The easiest way to do this is to change the values of the transfer coefficients to 0.

b) Run your model and keep track of the different reservoirs as a function of time. Describe and explain what you see. Were you surprised, or did you predict the result?

- Range = 1-1000 years, DT = 1/32 year

- Range = 1-1000, DT = 1/32 year
- After effectively removing the Land Biota reservoir and its inflows and outflows, the model still maintains a steady state, as predicted. Removing these components does not change any of the other reservoir values, as the values of the flows into and out of Land Biota were equal prior to the change and are equal (both zero) now.

9) The following questions come from the reading by Jahnke (1992). In recent years, many of the world's forests have been cut down and replaced with short-lived crops. What effect, if any, might this have on: (1) the P stored in the land biota reservoir, (2) the exchange rate of P between the land biota and the land reservoirs, and (3) the exchange rate between the land reservoir and the surface ocean?

- There’s not necessarily a right and wrong answer to this question. I’m most interested in having the students develop different hypotheses with which to experiment. Some potential consequences of deforestation might include:

1) Trees live longer and have more biomass than most crops, so it is possible that the amount of phosphorus stored in Land Biota would decrease if many of the world’s trees were replaced with crops. Since crops are more short-lived than trees, the residence time of P in Land Biota will be much lower in the model. Residence time is equal to one over the transfer coefficient, so the transfer coefficient governing the flux out of the Land Biota reservoir (decay) can be expected to increase in proportion to how much forest is replaced with farmland.

2) Trees tend to be more effective at preventing soil erosion than crops, so a change from trees to crops would probably result in an increase in the transfer coefficient for the flux from soils to ocean (weathering).

Modify your STELLA model to explore these questions and then explain your findings.

- There are many ways to alter the model to simulate geological, biological or chemical changes to the Earth’s phosphorus cycle. To explore the ideas brought up in question 9, a likely experiment would involve increasing the rate of decay, as well as increasing the rates of weathering, both insoluble and soluble. The graph below shows the results of an experiment in which the transfer coefficient for decay between Land Biota and Soils was increased by 30%, and the transfer coefficients for soluble and insoluble weathering were both increased by 10%.
- Range = 1-100 years, DT = 1/32 year

- These changes resulted in the Land Biota reservoir dropping dramatically over a very short period of time. The Soils reservoir increased dramatically at first due to the increased inflow from more rapidly decaying crops, but then began to decay back down towards its original level due to the increased weathering of soils produced by the lack of forests that limit erosion. All other reservoirs increase and approach new, higher steady state values.